

## Funkce INDEX

|             |        |
|-------------|--------|
| Anglicky    | INDEX  |
| Německy     | INDEX  |
| Francouzsky | INDEX  |
| Italsky     | INDICE |
| Španělsky   | INDICE |
| Rusky       | ИНДЕКС |

## Dostupnost ve verzích Excelu


|            |     |
|------------|-----|
| Excel 2003 | ANO |
| Excel 2007 | ANO |
| Excel 2010 | ANO |

## Definice

Funkce **INDEX** slouží k získání hodnoty (nebo skupiny hodnot) z oblasti na základě pořadového čísla.

## Zařazení

Funkci **INDEX** naleznete v kategorii **Vyhledávací funkce** společně s dalšími užitečnými funkcemi jako jsou **SVYHLEDAT**, **POZVYHLEDAT**, **POSUN**, **ZVOLIT**, **NEPŘÍMÝ.ODKAZ**.



## Syntaxe

**INDEX**(pole; číslo\_řádku; [čísló\_sloupce])

nebo

**INDEX**(oblast; číslo\_řádku; [čísló\_sloupce]; [čísló\_oblasti])

|   |   |   |  |   |
|---|---|---|--|---|
|   | A | B | C  | D |
| 1 |   |   |  |   |
| 2 |   |   | INDEX Fukce                                  |   |
| 3 |   |   |  |   |
| 4 |   |   | =INDEX(pole,                                 |   |
| 5 |   |   | čísló_řádku;[čísló_sloupce])                 |   |
| 6 |   |   | NEBO   |   |
| 7 |   |   | =INDEX(odkaz,                                |   |
| 8 |   |   | čísló_řádku;[čísló_sloupce];[čísló_oblasti]) |   |

**pole** = oblast buněk tvořících matici (tj. oblast, odkud bude hodnota vybrána). Jedná se o oblast buněk nebo maticovou konstantu – například {1;2;3;4}.

- Pokud argument pole určuje oblast pouze s jedním řádkem nebo sloupcem, může být příslušný argument řádek nebo sloupec vynechán.
- Pokud argument pole obsahuje více než jeden řádek a sloupec a je použit pouze jeden z argumentů řádek nebo sloupec, vrátí funkce INDEX celý řádek nebo sloupec.

**čísló\_řádku** = pořadové číslo řádku v matici

**čísló\_sloupce** = pořadové číslo sloupce v matici

**oblast** = odkazuje na oblast, ve které se bude hledat (může být zadáno více oblastí, pokud budou uzavřena do závorky)

- počet oblastí není omezený
- rozměry oblastí mohou být různé

**čísló\_oblasti** = číslo oblasti, která se má prohledávat. Číslování odpovídá pořadí, ve kterém jsou oblasti v argumentu zapsány.

Argument **pole** může být:

- odkazy na řádky nebo sloupce – A1:A100
- matice hodnot {1;2;3;4}
- odkazy na 2rozměrnou oblast – A1:K100
- 2 rozměrná matice hodnot {1\2\3;100\200\300;1000\2000\3000}
- definované názvy (DataProdeje) odkazující na oblast

Argument **oblast** může být:

- odkaz na libovolný počet oblastí s různě velkými rozměry uzavřený do kulatých závorek – (A1:G100; A200:G300;A400:I420)

Pomocí funkce INDEX lze také určit konkrétní řádek či sloupec dvourozměrné oblasti. Tím, že nevyplníte hodnotu pro sloupec (nebo použijete hodnotu 0), INDEX vyznačí oblast celého prvního řádku (pro vyznačení sloupce postupujte analogicky).

=INDEX(oblast;;1) - vybere první sloupec oblasti.

## K čemu můžete funkci INDEX použít?

- Nalezení hodnoty prodeje pro zadaný měsíc
- Nalezení hodnoty prodeje pro zadaný měsíc a produkt
- Nalezení hodnoty pro kombinaci kritérií
- Vytvoření dynamického seznamu hodnot
- Vytvoření abecedně seřazeného seznamu
- Nalezení nejbližší hodnoty
- Nalezení posledního výskytu hodnoty
- Vytvoření zdroje hodnot pro zobrazení dynamickým grafem

## Příklady

### A) pro začátečníky

### B) pro pokročilé

### C) pro znalce

## Časté chyby a problémy

Uživatelé použijí pro 2., 3. nebo 4. argument hodnotu mimo rozsah – například hodnotu 10 pro 2. argument nad oblastí A1:K9.

Při kopírování vzorce založeného na funkci INDEX dojde k posunutí odkazované oblasti a tedy k vzniku chyby. Předějit této chybě lze vytvořením absolutního odkazu pomocí znaku \$ - tedy místo odkazu A1:K100 v argumentu **pole** zadat odkaz ve formátu \$A\$1:\$K\$100.

## Rozdíly podle verzí Excelu

**Excel 97 - 2010** – funkce INDEX je implementována bez rozdílů

## Příklad A1: nalezení hodnoty prodeje pro zadaný měsíc

Základní způsob práce s funkcí INDEX je založen na výběru položky z jednoduchého seznamu hodnot (řádku, sloupce nebo matice). Pokud například buňka F2 obsahuje číslo měsíce, vrátí funkce INDEX pro tento měsíc příslušný objem prodeje. Funkce INDEX obsahuje v prvním argumentu 12 hodnot představující částky prodeje.

1) Hodnoty seznamu můžete zapsat přímo do prvního argumentu funkce:

=INDEX({311;185;95;172;239;205;181;64;128;206;274;135};F2)

2) Dosažené hodnoty prodeje mohou být pro první argument funkce také načteny z buněk v listu:

=INDEX(\$C\$2:\$C\$13;F2)

*Pozn. Řešený příklad je k dispozici v doprovodném sešitu na listu Cv01*

|   | AB | C             | D            | E | F            | G             | H |
|---|----|---------------|--------------|---|--------------|---------------|---|
| 1 |    | <b>Prodej</b> | <b>Měsíc</b> |   | <b>Měsíc</b> | <b>Prodej</b> |   |
| 2 |    | 311           | Leden        |   | 5            | 239           |   |
| 3 |    | 185           | Únor         |   |              |               |   |
| 4 |    | 95            | Březen       |   |              |               |   |
| 5 |    | 172           | Duben        |   |              |               |   |
| 6 |    | 239           | Květen       |   |              |               |   |
| 7 |    | 205           | Červen       |   |              |               |   |

Nalezení objemu prodeje v zadaném měsíci

## Příklad A2: určení hodnoty z průsečíku sloupce a řádku

Funkci INDEX nejčastěji používáme pro vytvoření dvourozměrného vyhledávání – nalezení hodnoty z průsečíku zadaného sloupce a řádku.

=INDEX(D3:H14;J3;K3) hodnota v buňce J3 určuje číslo řádku, hodnota v buňce K3 určuje číslo sloupce

*Pozn. Řešený příklad je k dispozici v doprovodném sešitu na listu Cv02*

L3      fx =INDEX(D3:H14;J3;K3)

|    | AB | C            | D              | E         | F         | G         | H         | I | J | K            | L              | M             | N            | O |
|----|----|--------------|----------------|-----------|-----------|-----------|-----------|---|---|--------------|----------------|---------------|--------------|---|
| 1  |    |              | <b>Produkt</b> |           |           |           |           |   |   |              |                |               |              |   |
| 2  |    | <b>Měsíc</b> | Produkt 1      | Produkt 2 | Produkt 3 | Produkt 4 | Produkt 5 |   |   | <b>Měsíc</b> | <b>Produkt</b> | <b>Prodej</b> | <b>Buňka</b> |   |
| 3  |    | Leden        | 149            | 217       | 304       | 130       | 159       |   |   | 5            | 3              | 230           | \$F\$7       |   |
| 4  |    | Únor         | 273            | 321       | 261       | 198       | 342       |   |   |              |                |               |              |   |
| 5  |    | Březen       | 236            | 94        | 297       | 312       | 69        |   |   |              |                |               |              |   |
| 6  |    | Duben        | 310            | 209       | 169       | 122       | 240       |   |   |              |                |               |              |   |
| 7  |    | Květen       | 87             | 71        | 230       | 270       | 208       |   |   |              |                |               |              |   |
| 8  |    | Červen       | 109            | 55        | 335       | 144       | 306       |   |   |              |                |               |              |   |
| 9  |    | Červenec     | 167            | 196       | 300       | 92        | 86        |   |   |              |                |               |              |   |
| 10 |    | Srpen        | 167            | 232       | 307       | 212       | 210       |   |   |              |                |               |              |   |
| 11 |    | Září         | 195            | 186       | 62        | 232       | 204       |   |   |              |                |               |              |   |
| 12 |    | Říjen        | 51             | 231       | 259       | 251       | 164       |   |   |              |                |               |              |   |
| 13 |    | Listopad     | 341            | 309       | 212       | 97        | 203       |   |   |              |                |               |              |   |
| 14 |    | Prosinec     | 67             | 196       | 98        | 149       | 110       |   |   |              |                |               |              |   |

Nalezení objemu prodeje v zadaném měsíci a zvýraznění buňky pomocí podmíněného formátování

Zobrazení průniku měsíce a produktu je nastaveno pomocí podmíněného formátování s využitím funkce POLÍČKO

## Příklad B1: nalezení hodnoty prodeje podle názvu měsíce

V praxi je funkce INDEX často kombinována s funkcí POZVYHLEDAT. Funkci POZVYHLEDAT používáme pro určení číselné hodnoty 2. a 3. argumentu funkce INDEX – tedy pro určení čísla řádku a sloupce.

Funkce POZVYHLEDAT pracuje jako "logický protihráč" funkce INDEX.

INDEX - vrací hodnotu buňky (nebo odkaz) určené v dané oblasti pořadovým číslem.

POZVYHLEDAT - vrací pořadové číslo buňky určené v dané oblasti hodnotou.

**=INDEX(\$C\$2:\$C\$8;POZVYHLEDAT(\$F\$2;\$D\$2:\$D\$8;0))**

POZVYHLEDAT(\$F\$2;\$D\$2:\$D\$8;0)

Funkce POZVYHLEDAT vyhledá hodnotu uloženou v buňce F2 („Červenec“) v oblasti D2:D8 (názvy měsíců). Výsledkem je hodnota **7** (měsíc „červenec“ je sedmá položka v prohledávané oblasti). Argument 0 určuje způsob hledání hodnoty – přesnou shodu mezi hledanou hodnotou a hodnotou v prohledávané oblasti.

Funkce INDEX(\$C\$2:\$C\$8;**7**) vrátí sedmou hodnotu z oblasti C2:C8, tedy hodnotu **181**.

|    |    | =INDEX(\$C\$2:\$C\$8;POZVYHLEDAT(\$F\$2;\$D\$2:\$D\$8;0)) |              |   |              |               |   |   |   |   |   |   |  |
|----|----|---|--------------|---|--------------|---------------|---|---|---|---|---|---|--|
|    | AB | C   | D            | E | F            | G             | H | I | J | K | L | M |  |
| 1  |    | <b>Prodej</b>   | <b>Měsíc</b> |   | <b>Měsíc</b> | <b>Prodej</b> |   |   |   |   |   |   |  |
| 2  |    | 311   | Leden        |   | Červenec     | 181           |   |   |   |   |   |   |  |
| 3  |    | 185   | Únor         |   |              |               |   |   |   |   |   |   |  |
| 4  |    | 95  | Březen       |   |              |               |   |   |   |   |   |   |  |
| 5  |    | 172   | Duben        |   |              |               |   |   |   |   |   |   |  |
| 6  |    | 239   | Květen       |   |              |               |   |   |   |   |   |   |  |
| 7  |    | 205   | Červen       |   |              |               |   |   |   |   |   |   |  |
| 8  |    | 181   | Červenec     |   |              |               |   |   |   |   |   |   |  |
| 9  |    | 64  | Srpen        |   |              |               |   |   |   |   |   |   |  |
| 10 |    | 128   | Září         |   |              |               |   |   |   |   |   |   |  |
| 11 |    | 206   | Říjen        |   |              |               |   |   |   |   |   |   |  |
| 12 |    | 274   | Listopad     |   |              |               |   |   |   |   |   |   |  |
| 13 |    | 135   | Prosinec     |   |              |               |   |   |   |   |   |   |  |

Nalezení objemu prodeje pro vybraný měsíc  
 Doplnění funkce INDEX o funkci POZVYHLEDAT  
 Výběr měsíce je realizován pomocí seznamu vytvořeném nástrojem **Ověření dat**

Pozn. Řešený příklad je k dispozici v doprovodném sešitu na listu **Cv04**

## Příklad B2: nalezení hodnoty prodeje podle názvu měsíce a produktu

V reálných případech je funkce INDEX doplněna dvojicí funkcí POZVYHLEDAT – jednou pro získání čísla řádku a podruhé pro získání čísla sloupce. Syntaxe zápisu vzorce je zcela identická k předchozímu příkladu.


M2      =INDEX(D4:H15;POZVYHLEDAT(K2;C4:C15;0);POZVYHLEDAT(L2;D3:H3;0))

|    | AB | C        | D         | E         | F         | G         | H         | I | J | K        | L         | M      |
|----|----|----------|-----------|-----------|-----------|-----------|-----------|---|---|----------|-----------|--------|
| 1  |    |          |           |           |           |           |           |   |   | Měsíc    | Produkt   | Prodej |
| 2  |    |          | Produkt   |           |           |           |           |   |   | Červenec | Produkt 3 | 300    |
| 3  |    | Měsíc    | Produkt 1 | Produkt 2 | Produkt 3 | Produkt 4 | Produkt 5 |   |   |          |           |        |
| 4  |    | Leden    | 149       | 217       | 304       | 130       | 159       |   |   |          |           |        |
| 5  |    | Únor     | 273       | 321       | 261       | 198       | 342       |   |   |          |           |        |
| 6  |    | Březen   | 236       | 94        | 297       | 312       | 69        |   |   |          |           |        |
| 7  |    | Duben    | 310       | 209       | 169       | 122       | 240       |   |   |          |           |        |
| 8  |    | Květen   | 87        | 71        | 230       | 270       | 208       |   |   |          |           |        |
| 9  |    | Červen   | 109       | 55        | 335       | 144       | 306       |   |   |          |           |        |
| 10 |    | Červenec | 167       | 196       | 300       | 92        | 86        |   |   |          |           |        |
| 11 |    | Srpen    | 167       | 232       | 307       | 212       | 210       |   |   |          |           |        |
| 12 |    | Září     | 195       | 186       | 62        | 232       | 204       |   |   |          |           |        |
| 13 |    | Říjen    | 51        | 231       | 259       | 251       | 164       |   |   |          |           |        |
| 14 |    | Listopad | 341       | 309       | 212       | 97        | 203       |   |   |          |           |        |
| 15 |    | Prosinec | 67        | 196       | 98        | 149       | 110       |   |   |          |           |        |

Pozn. Řešený příklad je k dispozici v doprovodném sešitu na listu **Cv05**

## Příklad B3: nalezení hodnoty z dílčí oblasti listu

Funkce INDEX je v Excelu k dispozici ve dvou variantách:



Pokud se rozhodnete použít variantu **INDEX(oblast;číslo\_řádku;[číslo\_sloupce];[číslo\_oblasti])**, která je bohužel v české lokalizaci označována poněkud zmatečně jako **INDEX(odkaz;číslo\_řádku;[číslo\_sloupce];[oblast])**, můžete zadat do prvního argumentu několik oblastí buněk (jejich počet není omezen) a to dokonce i oblasti s různým počtem řádků a sloupců.

Pomocí čtvrtého argumentu funkce INDEX pak můžete odkazovat jednotlivé oblasti a tedy vytvořit odkaz na požadovanou buňku (nebo sloupce či řádky) na základě 3 kritérií.

Odkazy na jednotlivé oblasti je nutné zadat do kulatých závorek, například takto:

=INDEX( (B3:C6;E3:F6;H3:I6) ;B10;C10;D10)

E10      fx      =INDEX((B3:C6;E3:F6;H3:I6);B10;C10;D10)

|    | A      | B        | C       | D       | E        | F       | G    | H      | I       | J    |
|----|--------|----------|---------|---------|----------|---------|------|--------|---------|------|
| 1  |        | Leden    |         |         | Únor     |         |      | Březen |         |      |
| 2  |        | výnosy   | náklady | zisk    | výnosy   | náklady | zisk | výnosy | náklady | zisk |
| 3  | východ | 935      | 137     | 798     | 528      | 222     | 306  | 724    | 144     | 580  |
| 4  | západ  | 926      | 294     | 632     | 944      | 387     | 557  | 788    | 308     | 480  |
| 5  | sever  | 696      | 431     | 265     | 814      | 249     | 565  | 758    | 166     | 592  |
| 6  | jih    | 849      | 415     | 434     | 988      | 223     | 765  | 575    | 399     | 176  |
| 7  |        |          |         |         |          |         |      |        |         |      |
| 8  |        |          |         |         |          |         |      |        |         |      |
| 9  |        | výnosy=1 |         |         |          |         |      |        |         |      |
|    |        | náklady= |         |         |          |         |      |        |         |      |
|    |        | oblast   | 2       | měsíc   | výsledek | buňka   |      |        |         |      |
| 10 |        | 3        | 2       | 2       | 249      | \$F\$5  |      |        |         |      |
| 11 |        | řádek    | sloupec | skupina |          |         |      |        |         |      |

Pozn. Řešený příklad je k dispozici v doprovodném sešitu na listu **Cv06**

## Příklad C1: vytvoření dynamické oblasti

Oblastí, ve které funkce INDEX nemá téměř konkurenci, je dynamické generování hodnot – například pro potřeby vytváření seznamů nebo hodnot pro uživatelsky modifikované grafy. Někteří uživatelé sice dávají přednost funkci POSUN, ale ta vzhledem ke své volatilitě je při práci s rozsáhlými datovými oblastmi pomalá – tento nedostatek se začíná projevovat především ve verzích Excel 2007|2010, které umožňují pracovat až s 2GB dat.

Pro vytváření dynamických oblastí pomocí funkce INDEX jsou používány následující 2 postupy:

1) oblast = buňka:INDEX

2) oblast = {INDEX}

### Typ oblast = buňka:INDEX

V příloženém sešitu je tímto způsobem vytvořena dynamická oblast pojmenovaná **MěsíceSeznam**, umožňující zobrazení položek oblasti v rozbalovacím seznamu:

Vzorec, který generuje tuto dynamickou oblast, má tvar **= $\$C\$1$ :INDEX( $\$C:\$C$ ;POČET2( $\$C:\$C$ ))**, kde

**C1** počáteční buňka oblasti

**POČET2( $\$C:\$C$ )** vrací číslo, které představuje počet vyplněných buněk ve sloupci C

**INDEX( $\$C:\$C$ ;POČET2( $\$C:\$C$ ))** vrací odkaz na poslední vyplněnou buňku ve sloupci C

Mezi první buňku sloupce (C1) a poslední vyplněnou buňku sloupce INDEX(\$C:\$C;POČET2('\$C:\$C)) je vložen operátor : (dvojtečka), který vytvoří odkaz na obdélníkovou oblast mezi těmito buňkami.

|   | A | B | C      | D     | E      | F      |
|---|---|---|--------|-------|--------|--------|
| 1 |   |   | Leden  |       |        | Měsíc  |
| 2 |   |   | Únor   |       |        | Březen |
| 3 |   |   | Březen | Leden |        |        |
| 4 |   |   | Duben  | Únor  | Březen |        |
| 5 |   |   | Květen | Duben | Květen |        |
| 6 |   |   |        |       |        |        |

Pozn. Řešený příklad je k dispozici v doprovodném sešitu na listu **Cv08**

## Typ oblast = {INDEX}

V tomto případě je možné zadat standardní syntaxi funkce INDEX - například INDEX (B32:D34;;2) – ale zadáme jej do buňky nebo oblasti buněk maticovým zápisem – tedy pomocí kombinace kláves

CTRL + SHIFT + ENTER.

Vizuálně jsou maticově zadané vzorce indikovány v řádku vzorců dvojicí složených závorek.

Tímto způsobem je zajištěno, že výsledek, který funkce INDEX vrací, bude matice hodnot nebo odkaz na oblast buněk.

|      |      |      |
|------|------|------|
| 1    | 2    | 3    |
| 100  | 200  | 300  |
| 1000 | 2000 | 3000 |

2202 {=SUMA(INDEX(B32:D34;;2))} ...vrací součet celého SLOUPCE č. 2  
 6000 {=SUMA(INDEX(B32:D34;3;))} ...vrací součet celého ŘÁDKU č. 3  
 6606 {=SUMA(INDEX(B32:D34;;;))} ...vrací součet celé OBLASTI

vzorce jsou zadány jako maticové - místo ENTER použijte CTRL+SHIFT+ENTER

## Příklad C2: použití funkce INDEX pro vytvoření dynamického grafu

Funkce INDEX je nejlepším řešením pro tvorbu dynamických grafů, které jsou základní součástí profesionálních reportů nebo firemních dashboardů.

Můžeme sice v některých případech použít i funkci POSUN, ale ta je řádově pomalejší ve své odezvě na požadavky uživatelů a může vést k neakceptovatelnému zpomalení generování grafů.


Hodnoty, které mají být v grafu zobrazeny na základě uživatelem zvoleného čísla sloupce, jsou generovány pomocí vzorce =INDEX(C5:F17;;D3) zapsaného maticovým způsobem (CTRL + SHIFT + ENTER) do pomocných buněk H5:H17, kde

**C5:F17** oblast hodnot, které požadují v grafu vykreslit

**D3** číslo sloupce, jehož hodnoty budou v grafu zobrazeny

Pokud marně hledáte oblast buněk, která slouží jako zdroj hodnot pro osu kategorií (tedy osa X), je tato oblast (B6:B17) zadaná do grafu staticky – tato oblast se nemění a statickým odkazem zrychlíme překreslování grafu.





Pozn. Řešený příklad je k dispozici v doprovodném sešitu na listu **Cv16**

Postup vytváření grafu naleznete jako videonávod prostřednictvím odkazu v sekci **Výukové video**

## Download – sešit obsahující cvičné příklady

Prezentované příklady (a několik dalších) jsou obsahem sešitu, který si můžete pro podrobné seznámení s funkcí INDEX stáhnout z webu DATASPECTRUM:

- [Verze pro Excel 97 – 2003](#)
- [Verze pro Excel 2007 | 2010](#)

## Výukové video

Pro rychlé seznámení s využitím funkce INDEX pro generování dynamického grafu si zobrazte výukové video:

YouTube: [ExcelAkademie - Dataspectrum: #012 | Funkce INDEX a dynamický graf](#)

Flash: [ExcelAkademie - Dataspectrum: #012 | Funkce INDEX a dynamický graf](#)

## Kurz – Mistrovství v práci s funkcemi Excelu

Seznamte se s pokročilými možnostmi práce s funkcemi Excelu

- Více než 200 řešených úloh.
- Příklady z praxe.
- Výuková videa pro snadné osvojení problematiky.
- Dosud nepublikované postupy a techniky práce.

Podrobné informace naleznete v [popisu kurzu](#).

Informace o dalších kurzech naleznete na adrese [Dataspectrum – Školení a kurzy](#).